Tableau Date Calculations Cheat Sheet

(Okay, it's more than one sheet)

How to use:

Step #1

Create the calculation that returns a boolean (True/False)

1. Keep the date field on the left-hand side of the equation
2. Place the comparison values on the right-hand side of the equation

Ex: Return Dates that are today

```
// is_today
[Date] = TODAY()
```

Step #2

Build a measure using the boolean

1. Use an IF statement
2. In the IF statement, use the date logic from Part 1
3. In the THEN statement, Use a row-level measure
4. (Optional) Aggregate the value

Ex: Aggregate Metric for Today

```
// Metric | Today
SUM(
  IF [is_today]
  THEN [Metric]
  END
)
```

Step #3

Compare the measure against other fields

1. Make sure all fields are aggregated

Ex: Compare Today and Yesterday

```
// Metric | Change Today vs.
Yesterday
[Metric | Today]
-
[Metric | Yesterday]
```

Part 1: Table of Contents

Most Recent Date `[Date | Max]`

Calculation Name

```
{MAX([Date])}
```

Calculation

Returns the most recent date in the dataset

Example:

- Data runs from January 1, 2022, through September 25, 2024.
- September, 25, 2024 will be used as the most recent date in the dataset

Year-to-Date or Current Year [CY]

```
DATETRUNC('year', [Date]) = DATETRUNC('year', [Date | Max])
```

Returns TRUE for dates within the current year.

Best used when data is available every day. If data is sparse or in real-time then replace [Date | Max] with TODAY().

```
YEAR([Date]) = YEAR([Date | Max]) - 1
```

Returns TRUE for dates for the prior year based on the maximum date in the dataset.

Alternative:
Returns TRUE for the current year of data.

Best used when data is available every day. If data is sparse or in real-time, then replace [Date | Max] with TODAY().

This technique returns the prior year for the first day of the new year.

```
DATEPART("year", [Date]) = DATEPART("year", [Date | Max]) -1
```



Prior-Year-to-Date [PYTD]

```
// PYTD - Based on Date of Year
(YEAR([Date]) = YEAR([Date | Max]) - 1)
AND
[Date] <= DATEADD('year', -1, [Date | Max])
```

Returns TRUE for dates from last year up to the current day of the year.

Best used when you are trying to match the exact day of the year. For example, if [Date | Max] is August 19, then the prior year will match to August 19th.



Prior-Year-to-Day [PYTDay]

```
// PYTD - Based on Day Number of Year
(YEAR([Date]) = YEAR([Date | Max]) - 1)
AND
DATEPART('dayofyear', [Date]) <= DATEPART('dayofyear', [Date | Max])
```

Returns TRUE for dates from last year up to the current day of the year.

This is an alternative to the prior year-to-date.

Matches the number of days into the year. Will return different dates on leap years or if using an ISO8601 date format.



Note the subtle difference in September 2023 between Prior-Year-to-Date and Prior-Year-to-Day.

Current Quarter [CQ]

```
DATETRUNC('quarter', [Date]) = DATETRUNC('quarter', [Date | Max])
```

Returns TRUE for dates within the current quarter

Best used when data is available every day. If data is sparse or in real-time then replace [Date | Max] with TODAY().

Last Full Quarter [LFQ]

```
DATETRUNC('quarter', [Date]) = DATEADD('quarter', -1, DATETRUNC('quarter', [Date | Max] + 1))
```

Returns TRUE for dates matching days in the previous quarter.

Use the calculation in your Previous Quarter to Date [PQTD] calculation.



Last Full Quarter of Previous Year [LQLY]

```
DATETRUNC('quarter', [Date]) = DATEADD('quarter', -5, DATETRUNC('quarter', [Date | Max] + 1))
```

Returns TRUE for dates matching days quarter from the previous year that matches the current years last full quarter.



```
DATEDIFF('day', DATETRUNC('quarter', [Date]), [Date])
```

Returns an integer representing the number of days a date is into a quarter.

Use the calculation in your Previous Quarter to Date [PQTD] and Quarter-to-Date of the Previous Year [QTDLY] calculations.



Days into current quarter [Current DIQ]

```
DATEDIFF('day', DATETRUNC('quarter', [Date | Max]), [Date | Max])
```

Returns an integer representing the number of days a date is into the current quarter.

Use the calculation in your Previous Quarter to Date [PQTD] and Quarter-to-Date of the Previous Year [QTDLY] calculations.



Previous Quarter to date [PQTD]

```
[LFQ]
AND
[Date | DIQ] <= [Current DIQ]
```

Returns TRUE for dates matching days in the previous quarter based on the day.



Same Days in Current Quarter [DIQ]

```
[Date | DIQ] <= [Current DIQ]
```

Returns TRUE for dates matching days in any quarter based on the baseline date used in [Current DIQ].

You will have to partition or filter by quarters to return expected values.

This date is either [Date | Max] or TODAY(), depending on how you have structured your dates.



Quarter-to-Date of Same Quarter in the Previous Year [QTDLY]

```
DATETRUNC('quarter', [Date]) = DATEADD('year', -1, DATETRUNC('quarter', [Date | Max]))
AND
[Date] <= DATEADD('year', -1, [Date | Max])
```

Returns TRUE for dates matching days for the quarter of the previous year based on the date into the current quarter and of the previous year.



Last 4 full quarters [L4FQ]

```
DATETRUNC('quarter', [Date]) >= DATEADD('quarter', -4, DATETRUNC('quarter', [Date | Max]+1))
AND
DATETRUNC('quarter', [Date]) < DATETRUNC('quarter', [Date | Max]+1)
```

Returns TRUE for dates matching days for last four complete quarters based on the date in [Date | Max].



Current Month or Month-to-Date [CM]

```
DATETRUNC('month', [Date]) = DATETRUNC('month', [Date | Max])
```

Returns TRUE for dates within the current month.

Best used when data is available every day. If data is sparse or in real-time then replace [Date | Max] with TODAY().

Last Full Month [LFM]

```
[Date] < DATETRUNC("month", [Date | Max] + 1)
AND
[Date] >= DATEADD("month", -1, DATETRUNC("month", [Date | Max] + 1))
```

Returns TRUE for the last full month of data.

Best used when data is available every day. If data is sparse or in real-time then replace [Date | Max] with TODAY().

Last Full Month of Previous Year [LMLY]

```
[Date] >= DATEADD("month", -13, DATETRUNC("month", [Date | Max] + 1))
```

Returns TRUE for the month from the prior year where there is currently the last full month of data.

Best used when data is available every day If data is sparse or in real-time then replace [Date | Max] with TODAY().



```
DATEDIFF('day', DATETRUNC('month', [Date]), [Date])
```

Returns an integer representing the number of days a date is into a month.

Use to calculation in your Previous Month to Date [PMTD] and Month-to-Date of Previous Year [MTDLY] calculations.



Days into current month [Current DIM]

```
DATEDIFF('day', DATETRUNC('month', [Date | Max]), [Date | Max])
```

Returns an integer representing the number of days a date is into the current month.

Use to calculation in your Previous Month to Date [PMTD] and Month-to-Date of Previous Year [MTDLY] calculations.



Previous Month to date [PMTD]

```
[LFM]
AND
[Date | DIM] <= [Current DIM]
```

Returns TRUE for dates matching days in the previous month based on the day.

Same Days in Current Month [SDCM]

```
[Date | DIQ] <= [Current DIQ]
```

Returns TRUE for dates matching days in any month based on the baseline date used in [Current DIM].

This date is either [Date | Max] or TODAY() depending on how you have structured your dates.



Month-to-Date of Same Month in the Previous Year [MTDLY]

```
DATETRUNC('month', [Date]) = DATEADD('year', -1, DATETRUNC('month', [Date | Max]))
AND
[Date] <= DATEADD('year', -1, [Date | Max])
```

Returns TRUE for dates matching days for the month of the previous year based on the date into the current month and of the previous year.



Last 12 Full Months [L12FM]

```
DATETRUNC('month', [Date]) >= DATEADD('month', -12, DATETRUNC('month', [Date | Max]+1))
AND
DATETRUNC('month', [Date]) < DATETRUNC('month', [Date | Max]+1)
```

Returns TRUE for dates matching days for last twelve complete months based on the date in [Date | Max].



Last 12 Months [L12M]

```
DATETRUNC('month', [Date]) >= DATEADD('month', -11, DATETRUNC('month', [Date | Max]+1))
AND
DATETRUNC('month', [Date]) <= [Date | Max]
```

Returns TRUE for dates matching days for last eleven full months and the current incomplete month based on the date in [Date | Max].

Month of Quarter [MOQ]

```
((DATEPART('month', [Date]) -1) % 3) + 1
```

Returns an integer for the number of the month in the quarter. Either 1, 2, or 3.



Week-to-Date or Current Week [CW]

```
DATETRUNC('week', [Date]) = DATETRUNC('week', [Date | Max])
```

Returns TRUE for dates within the current week.

Best used when data is available every day. If data is sparse or in real-time, then replace [Date | Max] with TODAY().



Last Full Week [LFW]

```
DATETRUNC('week', [Date]) = DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1))
```

Returns TRUE for dates for the last full week completed.

Same Week of Current Date for the Previous Year [LWLY]

```
DATETRUNC('week', [Date]) = DATETRUNC('week', DATEADD('year', -1, DATEADD('week', -1, DATETRUNC('week', [Date | Max]
+1))))
```

Returns TRUE for dates one year prior to the last fully completed week based on day of the week.

Same Week Number of Last Full Week for the Previous Year [LWNumLY]

```
DATEPART('week', [Date]) = DATEPART('week', DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1)))
AND
DATEPART('year', [Date]) = DATEPART('year', DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1))) -1
```

Returns TRUE for dates one year prior to the last fully completed week based on week number.



Last 52 Full Weeks [L52]

```
DATETRUNC('week', [Date]) <= DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1))
AND
DATETRUNC('week', [Date]) >= DATEADD('week', -52, DATETRUNC('week', [Date | Max] +1))
```

Returns TRUE for dates for the last 52 fully completed weeks.



Last 13 Full Weeks [L13W]

```
DATETRUNC('week', [Date]) <= DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1))
AND
DATETRUNC('week', [Date]) >= DATEADD('week', -13, DATETRUNC('week', [Date | Max] +1))
```

Returns TRUE for dates for the last 13 fully completed weeks.



Previous 13 Full Weeks [P13W]

```
DATETRUNC('week', [Date]) <= DATEADD('week', -14, DATETRUNC('week', [Date | Max] +1))
AND
DATETRUNC('week', [Date]) >= DATEADD('week', -26, DATETRUNC('week', [Date | Max] +1))
```

Returns TRUE for dates for the previous 13 completed weeks prior to the 13 most recent completed weeks (weeks 14-26).

Last 13 Full Weeks for the Prior Year [L13WLY]

```
DATETRUNC('week', [Date]) <= DATEADD('year', -1, DATEADD('week', -1, DATETRUNC('week', [Date | Max] +1)))
AND
DATETRUNC('week', [Date]) >= DATEADD('year', -1, DATEADD('week', -13, DATETRUNC('week', [Date | Max] +1)))
```

Returns TRUE for dates one year prior to the last 13 fully completed weeks.



Prior Year-to-Last-Full Week [WTDLY]

```
[Date] < DATETRUNC('week', DATEADD('year', -1, DATETRUNC('week', [Date | Max] +1)))
AND
YEAR([Date]) = YEAR(DATEADD('year', -1, [Date | Max] + 1))
```

Returns TRUE for dates from previous year through the equivalent of the last fully completed week of this year.



Week Number of Each Month [Week No | Month]

```
DATEPART('week', [Date]) - {FIXED DATETRUNC('month', [Date]) : MIN(DATEPART('week', [Date]))} + 1
```

Returns an integer based on the week number of the month. The first week–partial or whole–starts at 1 and counts up. Values range from 1 to 6.

Last Day or Current Day [CD]

```
DATETRUNC('day', [Date]) = DATETRUNC('day', [Date | Max])
```

Returns TRUE for dates within the current week.

Best used when data is available every day. If data is sparse or in real-time, then replace [Date | Max] with TODAY().



One week prior of Last Day [LDLY]

```
[Date] = DATEADD('week', -1, DATETRUNC('day', [Date | Max]))
```

Returns TRUE for the day that is one week prior to the last day reported



Weekdays Only `[is_weekday]`

```
DATEPART('weekday', [Date]) IN(2, 3, 4, 5, 6)
```

Returns TRUE for dates where the weekday is Monday, Tuesday, Wednesday, Thursday, or Friday.



Weekends Only `[is_weekend]`

```
DATEPART('weekday', [Date]) IN(1, 7)
```

Returns TRUE for dates where the weekday is Saturday or Sunday.



Most Recent 30 Days [L30D]

```
DATEDIFF('day', [Date], [Date | Max]) < 30
```

Returns TRUE for dates within last thirty days of last reported date.

Previous 30 Days Prior [P30D]

```
DATEDIFF('day', [Date], [Date | Max]) >= 30
AND
DATEDIFF('day', [Date], [Date | Max]) < 60
```

Returns TRUE for dates within days 31 to 60 of most recent date.

Third Thursday of the Month [3ThM]

```
[Date] = DATEADD('week', 2, {FIXED DATETRUNC('month', [Date]) : MIN(IF DATEPART('weekday', [Date]) = 5 THEN [Date]
END)})
```

Returns TRUE for dates that are the third Thursday of a given month. Values can be edited to return any single week number-weekday pairing.



Decade [Decade]

```
INT(YEAR([Date])/10) * 10
```

Returns Integer of a decade of a year of a date

Create Date [YM]

```
MAKEDATE([Year], [Month], 1)
```

Returns date when given only given two integers representing year and month, respectively.

Table Calculations: Last 13 Weeks

```
// Last 13 Values
LAST() < 13
```

Returns TRUE for the last 13 values to the right-hand side of the chart.

Similarly, the Last 12 months can be done with `LAST() < 12` .

Instructions
Place a date field at the week level on the columns shelf.

Add a measure to the rows shelf.

Create the field [Last 13 Values] and place it on the filters shelf. Select True.



```
// Last 12 Values
LAST() < 12
```

Returns TRUE for the last 12 values to the right-hand side of the chart.

```
// Measure | % Change | MoM
SUM([Measure]) - LOOKUP(ZN(SUM([Measure])), -12) / ABS(LOOKUP(ZN(SUM([Measure])), -12))
```

Returns float as a percentage comparing current values to the value twelve marks prior–in this case the same month of the previous year.

Instructions
Place a date field at the week level on the columns shelf.

Add a measure [Measure | % Change | MoM] to the rows shelf.

Place [Last 12 Values] filters shelf. Select True.



Enjoying the Tableau Date Calculations Cheat Sheet?

Explore our Tableau services and case studies to see how we've helped other companies achieve success with their data.

Learn more at phdata.io